

## 29. SĄLYGINĖ FUNKCIJA IF



Skaiciuoklė moka atlikti įvairius veiksmus: sudėti, atimti, dauginti, dalyti, kelti laipsniu ir traukti iš skaičiaus šaknį. Tačiau kartais, atliekant veiksmus su duomenimis, atsiranda pranešimas apie klaidą (pvz., #NUM! arba #DIV/0!). Taip yra todėl, kad kai kuriuos veiksmus galima atlikti ne su visais skaičiais. Pavyzdžiui, negalima ištraukti šaknies iš neigiamo skaičiaus:

	A	B	C	D	E
1	Kvadratinės šaknies traukimas iš skaičiaus a				
2	Skaičius a	4	4,5	-8	9
3	Kvadratinė šaknis iš a	2	2,12132	#NUM!	3

Klaidų galima išvengti naudojant sąlyginę funkciją (IF). Ji leidžia atlikti vieną iš dviejų veiksmų atsižvelgiant į tai, ar nurodyta sąlyga tenkinama, ar netenkinama.

**Bendru atveju funkcija IF rašoma taip:**

**IF(sąlyga; reikšmė\_kai\_sąlyga\_tenkinama; reikšmė\_kai\_sąlyga\_netenkinama)**

Pasinaudokime sąlygine funkcija kvadratinei šakniai iš skaičiaus a apskaičiuoti. Iš pradžių funkcija IF turi patikrinti, ar skaičius neneigiamas, ir tik tada traukti šaknį. Jei skaičius neigiamas, reikia nurodyti, kokį pranešimą išvesti, pavyzdžiui, *Nėra prasmės*.

Langelyje B3 rašoma formulė. Funkcijos IF skliaustuose pirmiausia suformuluojama sąlyga, t. y. klausimas, ar langelyje B2 esantis skaičius, iš kurio norima traukti šaknį, yra neneigiamas: **B2>=0**. (Skaiciuoklė nesupranta ženklų >, < ir ≠. Todėl rašant sąlygą daugiau arba lygu, mažiau arba lygu, nelygu reikalingi du ženklai: >=, <=, <>.)

Po sąlygos rašomas kabliataškis ir nurodoma, ką įrašyti į langelį B3, jei sąlyga tenkinama, t. y. ištrauktą šaknį iš langelyje B2 esančio skaičiaus: **B2^(1/2)**.

Tada vėl rašomas kabliataškis ir nurodoma, ką įrašyti į langelį B3, kai langelyje B2 esantis skaičius yra neigiamas, pavyzdžiui, *Nėra prasmės*. (Tekstiniai pranešimai formulėje rašomi tarp paprastųjų kabučių.)

Funkcijos IF skliaustai uždaromi ir spustelimas klavišas **Įvesti (Enter)**.

Į langelį B3 įrašyta formulė atrodo taip:

**=IF(B2>=0;B2^(1/2);"Nėra prasmės")**

Sukurtą formulę reikia nukopijuoti į kitus langelius (C3:E3).

Tas pats kvadratinės šaknies iš a reikšmės gautume, jei sąlyginę funkciją užrašytume kitaip – klausime, ar langelyje esantis skaičius yra neigiamas: **B2<0**. Jei ši sąlyga tenkinama, langelyje B3 nurodytume skaičiuoklei įrašyti *Nėra prasmės*. Visais kitais atvejais langelyje B3 skaičiuoklė turi įrašyti reikšmę, gautą iš langelyje B2 esančio skaičiaus ištraukus šaknį: **B2^(1/2)**.

B3	=IF(B2>=0;B2^(1/2);"Nėra prasmės")				
	A	B	C	D	E
1	Kvadratinės šaknies traukimas iš skaičiaus a				
2	Skaičius a	4	4,5	-8	9
3	Kvadratinė šaknis iš a	2	2,12132	Nėra prasmės	3

B3	=IF(B2<0;"Nėra prasmės";B2^(1/2))				
	A	B	C	D	E
1	Kvadratinės šaknies traukimas iš skaičiaus a				
2	Skaičius a	4	4,5	-8	9
3	Kvadratinė šaknis iš a	2	2,12132	Nėra prasmės	3

	A	B	C	D
1	Trikampio kraštinės			Trikampio rūšis
2	a	b	c	
3	3	4	5	
4	2	2	2	
5	15	8	17	
6	8	6	10	
7	5	5	6	

Sąlygoje gali būti naudojami ir reiškiniai. Tarkime, reikia patikrinti, ar trikampis yra statusis, kai žinomos jo kraštinės a, b ir c. Skaiciuoklė parenkime lentelę, pavaizduotą kairėje.

Norint nustatyti, ar trikampis statusis, reikia pasinaudoti Pitagoro teorema. Jei statinių a ir b kvadratų suma lygi įžambinės c kvadratu, tai trikampis statusis. Langelyje D3 įrašome atitinkamą formulę ir ją kopijuojame į kitus stulpelio D langelius.

	A	B	C	D
1	Trikampio kraštinės			Trikampio rūšis
2	a	b	c	
3	3	4	5	=IF(A3^2+B3^2=C3^2;"Statusis";"Nestatusis")
4	2	2	2	=IF(A4^2+B4^2=C4^2;"Statusis";"Nestatusis")
5	15	8	17	=IF(A5^2+B5^2=C5^2;"Statusis";"Nestatusis")
6	8	6	10	=IF(A6^2+B6^2=C6^2;"Statusis";"Nestatusis")
7	5	5	6	=IF(A7^2+B7^2=C7^2;"Statusis";"Nestatusis")

	A	B	C	D
1	Trikampio kraštinės			Trikampio rūšis
2	a	b	c	
3	3	4	5	Statusis
4	2	2	2	Nestatusis
5	15	8	17	Statusis
6	8	6	10	Statusis
7	5	5	6	Nestatusis

Sąlygą galima užrašyti ir kitaip:  $c^2 = a^2 + b^2$ . Tada langelyje D3 formulė būtų tokia: **=IF(C3^2=A3^2+B3^2;"Statusis";"Nestatusis")**.

Bet kuriuo atveju skaičiuoklė pirmiausia apskaičiuoja sąlygoje nurodytų reiškinių reikšmes ir tik po to jas lygina. Jei įrašytas lygybės ženklas (kaip šiuo atveju), tai skaičiuoklė patikrina, ar reiškiniai abiejose lygybės pusėse lygūs. Jei taip, į langelį įrašomas žodis *Statusis*, jei ne – *Nestatusis*.

	A	B	C	D
1	Trikampio kraštinės			Trikampio rūšis
2	a	b	c	
3	3	4	5	Statusis
4	2	2	2	FALSE
5	15	8	17	Statusis
6	8	6	10	Statusis
7	5	5	6	FALSE

Jei sąlyga netenkinama ir nieko nenurodyta skaičiuoklei, ką tuo atveju daryti, pavyzdžiui, **=IF(A3^2+B3^2=C3^2;"Statusis")**, tai langelyje atsiranda žodis *FALSE* (netiesa).

Jie norime, kad vietoj žodžio *FALSE* langelyje būtų rašomas nulis, tai formulėje po žodžio *Statusis* reikia rašyti kabliataškį:

**=IF(A3^2+B3^2=C3^2;"Statusis";)**

Norint, kad skaičiuoklė nieko neįrašytų, geriausiai padėti dvi paprastas kabutes: **=IF(A3^2+B3^2=C3^2;"Statusis";"**

	A	B	C	D
1	Trikampio kraštinės			Trikampio rūšis
2	a	b	c	
3	3	4	5	Statusis
4	2	2	2	
5	15	8	17	Statusis
6	8	6	10	Statusis
7	5	5	6	

	A	B	C	D
1	Trikampio kraštinės			Trikampio rūšis
2	a	b	c	
3	3	4	5	TRUE
4	2	2	2	FALSE
5	15	8	17	TRUE
6	8	6	10	TRUE
7	5	5	6	FALSE

Užuot rašius visą sąlyginę funkciją, galima langeliuose nurodyti tik **=A3^2+B3^2=C3^2**. Tuomet lentelė būtų tokia, kaip pateikta kairėje.

Sąlyginėje funkcijoje gali būti naudojamos kitos sąlyginės funkcijos. Tai jau sudėtingesni atvejai.